// Unity built-in shader source. Copyright (c) 2016 Unity Technologies. MIT license (see license.txt)

Shader "UI/GpuImage"
{
    Properties
    {
        _Main1Tex ("Sprite Texture", 2D) = "white" {}
        _Main2Tex ("Sprite Texture", 2D) = "white" {}
        _Main3Tex ("Sprite Texture", 2D) = "white" {}
        _Main4Tex ("Sprite Texture", 2D) = "white" {}
        _Color ("Tint", Color) = (1,1,1,1)
        _HeadRect("headRect",Vector) = (0.2,0.2,0.2,0.2)
        
        _StencilComp ("Stencil Comparison", Float) = 8
        _Stencil ("Stencil ID", Float) = 0
        _StencilOp ("Stencil Operation", Float) = 0
        _StencilWriteMask ("Stencil Write Mask", Float) = 255
        _StencilReadMask ("Stencil Read Mask", Float) = 255

        _ColorMask ("Color Mask", Float) = 15

        [Toggle(UNITY_UI_ALPHACLIP)] _UseUIAlphaClip ("Use Alpha Clip", Float) = 0
    }

    SubShader
    {
        Tags
        {
            "Queue"="Transparent"
            "IgnoreProjector"="True"
            "RenderType"="Transparent"
            "PreviewType"="Plane"
            "CanUseSpriteAtlas"="True"
        }

        Stencil
        {
            Ref [_Stencil]
            Comp [_StencilComp]
            Pass [_StencilOp]
            ReadMask [_StencilReadMask]
            WriteMask [_StencilWriteMask]
        }

        Cull Off
        Lighting Off
        ZWrite Off
        ZTest [unity_GUIZTestMode]
        Blend One OneMinusSrcAlpha
        ColorMask [_ColorMask]

        Pass
        {
            Name "Default"
        CGPROGRAM
            #pragma vertex vert
            #pragma fragment frag
            #pragma target 2.0

            #include "UnityCG.cginc"
            #include "UnityUI.cginc"

            #pragma multi_compile_local _ UNITY_UI_CLIP_RECT
            #pragma multi_compile_local _ UNITY_UI_ALPHACLIP
            #pragma multi_compile_instancing

            struct appdata_t
            {
                float4 vertex   : POSITION;
                float4 color    : COLOR;
                float2 texcoord : TEXCOORD0;
                UNITY_VERTEX_INPUT_INSTANCE_ID
            };

            struct v2f
            {
                float4 vertex   : SV_POSITION;
                fixed4 color    : COLOR;
                float2 texcoord  : TEXCOORD0;
                float4 worldPosition : TEXCOORD1;
                float4  mask : TEXCOORD2;
                UNITY_VERTEX_OUTPUT_STEREO
                UNITY_VERTEX_INPUT_INSTANCE_ID
            };

            sampler2D _Main1Tex;
            sampler2D _Main2Tex;
            sampler2D _Main3Tex;
            sampler2D _Main4Tex;
            float4 _Main1Tex_ST;
            float4 _Main2Tex_ST;
            float4 _Main3Tex_ST;
            float4 _Main4Tex_ST;
            fixed4 _Color;
            fixed4 _TextureSampleAdd;
            float4 _ClipRect;
            float _UIMaskSoftnessX;
            float _UIMaskSoftnessY;
            float4 _HeadRect;
            int _UIVertexColorAlwaysGammaSpace;

            UNITY_INSTANCING_BUFFER_START(Props)
	        UNITY_DEFINE_INSTANCED_PROP(float, imgIndex)
                #define imgIndex_arr Props
            UNITY_DEFINE_INSTANCED_PROP(float4, headUV)
                #define headUV_arr Props
            UNITY_INSTANCING_BUFFER_END(Props)
        
            v2f vert(appdata_t v)
            {
                v2f OUT;
                UNITY_SETUP_INSTANCE_ID(v);
                UNITY_TRANSFER_INSTANCE_ID(v,OUT);
                UNITY_INITIALIZE_VERTEX_OUTPUT_STEREO(OUT);
                float4 vPosition = UnityObjectToClipPos(v.vertex);
                OUT.worldPosition = v.vertex;
                OUT.vertex = vPosition;

                float2 pixelSize = vPosition.w;
                pixelSize /= float2(1, 1) * abs(mul((float2x2)UNITY_MATRIX_P, _ScreenParams.xy));

                float4 clampedRect = clamp(_ClipRect, -2e10, 2e10);
                float2 maskUV = (v.vertex.xy - clampedRect.xy) / (clampedRect.zw - clampedRect.xy);

                float index = UNITY_ACCESS_INSTANCED_PROP(imgIndex_arr, imgIndex);
                if(index >=3)
                {
                   OUT.texcoord = TRANSFORM_TEX(v.texcoord.xy, _Main3Tex);   
                }
                else if(index >=2)
                {
                    OUT.texcoord = TRANSFORM_TEX(v.texcoord.xy, _Main2Tex);   
                }
                else
                {
                     OUT.texcoord = TRANSFORM_TEX(v.texcoord.xy, _Main1Tex);   
                }
                OUT.mask = float4(v.vertex.xy * 2 - clampedRect.xy - clampedRect.zw, 0.25 / (0.25 * half2(_UIMaskSoftnessX, _UIMaskSoftnessY) + abs(pixelSize.xy)));

                if (_UIVertexColorAlwaysGammaSpace)
                {
                    if(!IsGammaSpace())
                    {
                        v.color.rgb = UIGammaToLinear(v.color.rgb);
                    }
                }

                OUT.color = v.color * _Color;
                return OUT;
            }

            fixed4 frag(v2f IN) : SV_Target
            {
                //Round up the alpha color coming from the interpolator (to 1.0/256.0 steps)
                //The incoming alpha could have numerical instability, which makes it very sensible to
                //HDR color transparency blend, when it blends with the world's texture.
                const half alphaPrecision = half(0xff);
                const half invAlphaPrecision = half(1.0/alphaPrecision);
                IN.color.a = round(IN.color.a * alphaPrecision)*invAlphaPrecision;


                half4 color;
                UNITY_SETUP_INSTANCE_ID(IN);
                float index = UNITY_ACCESS_INSTANCED_PROP(imgIndex_arr, imgIndex);
                if(index >=3)
                {
                  color = IN.color * (tex2D(_Main3Tex, IN.texcoord) + _TextureSampleAdd);
                }
                else if(index >=2)
                {
                    color = IN.color * (tex2D(_Main2Tex, IN.texcoord) + _TextureSampleAdd);  
                }
                else
                {
                    color = IN.color * (tex2D(_Main1Tex, IN.texcoord) + _TextureSampleAdd);  
                }
                
                bool head = IN.texcoord.x >= _HeadRect.x && IN.texcoord.x <= _HeadRect.x + _HeadRect.z && IN.texcoord.y >= _HeadRect.y && IN.texcoord.y <= _HeadRect.y + _HeadRect.w;
                if(head)
                {
                    float4 rect = UNITY_ACCESS_INSTANCED_PROP(headUV_arr, headUV);
                    float headU = (IN.texcoord.x - _HeadRect.x) /_HeadRect.z;
                    float headV = (IN.texcoord.y- _HeadRect.y) /_HeadRect.w;
                    float2 uv =  float2(headU,headV);
                    fixed4 headColor = tex2D(_Main4Tex, rect.xy + uv*rect.zw);
                    if(headColor.a >0.1)
                    {
                         color = headColor;   
                    }
                }
                #ifdef UNITY_UI_CLIP_RECT
                half2 m = saturate((_ClipRect.zw - _ClipRect.xy - abs(IN.mask.xy)) * IN.mask.zw);
                color.a *= m.x * m.y;
                #endif

                #ifdef UNITY_UI_ALPHACLIP
                clip (color.a - 0.001);
                #endif

                color.rgb *= color.a;
                return color;
            }
        ENDCG
        }
    }
}
